

## Fiche n° 5 - Explorer le contenu d'une table et afficher une liste de valeur

Le script suivant permet de lire la table biblio, d'extraire le titre, l'auteur, l'éditeur et la date de lecture des documents.

Un test permet de sélectionner uniquement les documents lus dans les 365 jours précédant la date du traitement.

```
; Liste des documents lus dans les 365 derniers jours
var
    tc TCursor
    dynView DynArray[] AnyType
endVar
; on met la date du jour - 365 jours dans une variable nommé datedujour
datedujour = today()-365
; on extrait de la date du jour, l'année, le mois et le jour
Annee = strval(datedujour.year())
Mois = strval(datedujour.month())
Jour = strval(datedujour.day())
    tc.open("biblio.db")
    compte = 0
    scan tc:
        if tc.date1 > today()-365
            then
                compte = compte + 1
                laref = strval(int(tc.reference))
                dynView[laref] = tc.titre + " / "+tc.auteur+ " /
"+tc.editeur+ " / "+ tc.date1
            endif
        endscan

    dynView.view(strval(int(compte))+" Document(s) lus depuis le
"+strval(Jour)+"/"+strval(Mois)+"/"+strval(annee))
    tc.close()
```

**nb :** *tc.date1* correspond à la rubrique Date 1 située dans l'onglet « Complément » de l'écran principal. Dans notre base de données exemple, cette rubrique personnalisable est utilisée pour renseigner la date de lecture.

Dans cet exemple nous avons utilisé 4 nouvelles instructions :

- Today()
- DynArray
- Strval
- Int

**Today()**

Renvoie la date du jour.

**Syntaxe :**

today ( ) Date

**Description :**

today donne la date du jour, selon l'horloge/le calendrier du système.

**Exemple :**

```
; affiche la date actuelle
msgInfo("Nous sommes aujourd'hui le :", today())
```

**DynArray :**

Un Array (tableau) contient des valeurs (appelées articles ou éléments) dans des cellules, de la même façon que des casiers de tri postal contiennent des lettres. Un tableau ObjectPAL est mono-dimension et peut être comparé à une rangée de casiers, comprenant un article par casier.

Pour utiliser des tableaux dans les méthodes, il faut les déclarer en leur attribuant un nom, une taille (le nombre d'articles) et un type de donnée (exemple : string, number ou anytype) pour les articles.

Dans l'exemple suivant, le script crée un tableau dynamique puis affiche dans une boîte de dialogue son contenu trié.

```
var
  td      DynArray[] String
endVar

td ["un"]   = "premier"
td ["deux"] = "deuxième"
td ["trois"] = "troisième"
td.view("Ce tableau contient :")
; affiche :
; Ce tableau contient :
; un      Premier
; trois   Troisième
; deux    Deuxième
```

**Strval :**

Convertit une valeur en chaîne.

**Syntaxe :**

strVal (valeur)

**Description :**

strVal convertit la valeur en chaîne. La valeur est de n'importe quel type représenté par AnyType.

**Int :**

Transforme une valeur en nombre entier.

**Syntaxe :**

int (valeur)

**Description :**

int transforme (convertit) l'expression numérique valeur en nombre entier. Dans le cas d'une conversion à partir d'un type plus précis (Number par exemple), il peut y avoir perte de précision.